

# Supplementary Material 1

## Example NONMEM code used in the model-integrated bioequivalence method

### 1. Example NONMEM code for estimation step

```
$PROBLEM      PK estimation
$INPUT        REP=DROP ID TIME AMT EVID DV TRT SEQ PER
$DATA         be_data.csv IGNORE=@
$SUBROUTINE   ADVAN2 TRANS2
$PK
FTRT = 1
IF (TRT.EQ.2) FTRT = THETA(5)

KATRT = 1
IF (TRT.EQ.2) KATRT = THETA(6)

FSEQ = 1
IF (SEQ.EQ.2) FSEQ = THETA(7)

KASEQ = 1
IF (SEQ.EQ.2) KASEQ = THETA(8)

FPER = 1
IF (PER.EQ.2) FPER = THETA(9)

KAPER = 1
IF (PER.EQ.2) KAPER = THETA(10)

IF (PER.EQ.1) IOVF = ETA(4)
IF (PER.EQ.2) IOVF = ETA(5)

CL = THETA(1) * EXP(ETA(1))
V = THETA(2) * EXP(ETA(2))
F1 = THETA(3) * EXP(IOVF) * FTRT * FSEQ * FPER
KA = THETA(4) * EXP(ETA(3)) * KATRT * KASEQ * KAPER

$ERROR
IPRED = A(2)/V
IRES = DV-IPRED
PROP = SQRT(SIGMA(1,1))*IPRED
W = PROP
IF (W.EQ.0) W=1
IWRES = IRES/W
Y = IPRED + IPRED*EPS(1)

$THETA
(0,46) ; 1.CL
(0,400) ; 2.V
1 FIX. ; 3.F
(0, 1.2) ; 4.KA
(0,1.25) ; 5.FTRT
(0, 1.1) ; 6. KATRT
(0, 1.1) ; 7.FSEQ
(0, 1.1) ; 8. KASEQ
(0, 1.1) ; 9.FPER
```

```

(0, 1.1) ; 10. KAPER

$OMEGA
0.03. ; 1. CL
0.02 ; 2. V
0.03 ; 3. KA
$OMEGA BLOCK(1) 0.0025 ;IOVF
$OMEGA BLOCK(1) SAME ;IOVF

$SIGMA
0.01 ;PROP

;;; estimation step
$ESTIMATION PRINT=1 MAXEVAL=9999 METHOD=1 INTER NOABORT SADDLE_RESET=1

;;; uncertainty: covariance matrix
$COVARIANCE UNCONDITIONAL

;;; uncertainty: SIR method
$COV SIRSAMPLE=2000 SIRNITER=6 SIRDF=30 IACCEPT=0.4 FILE=output_sir.ext

;;; uncertainty: bootstrap method
$SIML (101) BOOTSTRAP=-1 SUBP=1000 STRAT=SEQ

```

## 2. Example NONMEM code for the simulation step of a more complicated model (non-linear elimination) defined using ordinary differential equations

The input dataset (simdata.csv) includes:

- Dosing records for the 2-period BE study design with the doses the same as in the original BE data
- Observation records
  - at the last sampling time of the original BE dataset to calculate  $AUC_{last}$
  - at an extended time point (5 times of the last sampling time) to calculate  $AUC_{inf}$
- treatment covariate (TRT) of 1 and 2 for each period, respectively
- Period covariate (PER) and sequence covariate (SEQ) set to 1 for both periods

$AUC_{inf}$  is extracted from the output table file sim\_result.txt at the extended time point and  $AUC_{last}$  is extracted from the value reported at the last sampling time of the original BE dataset.

```

$PROBLEM PK simulation
$INPUT ID TIME DV AMT EVID CMT PER SEQ TRT DOSE
$DATA simdata.csv IGNORE=@
$SUBROUTINE ADVAN13 TOL=6
$MODEL
COMP=(DEPOT) ;1. Depot compartment
COMP=(CENTRAL) ;2. Central compartment
COMP=(AUC NODOSE) ;3. AUC calculation

$PK

```

```

IF (NEWIND.NE.2) THEN
  TDOS = 0
  TAD = 0
ENDIF

IF (EVID.EQ.1.OR.EVID.EQ.4) THEN
  TDOS = TIME
ENDIF

TAD = TIME - TDOS

FTRT = 1
IF (TRT.EQ.2) FTRT = THETA(5)

KATRT = 1
IF (TRT.EQ.2) KATRT = THETA(6)

IF (PER.EQ.1) IOVF = ETA(4)
IF (PER.EQ.2) IOVF = ETA(5)

VM = THETA(1) * EXP(ETA(1))
V = THETA(2) * EXP(ETA(2))
F1 = THETA(3) * EXP(IOVF) * FTRT
KA = THETA(4) * EXP(ETA(3)) * KATRT
KM = THETA(11)

AUCI = F1*DOSE/CL ;for linear PK

$DES
DADT(1) = - KA*A(1)
DADT(2) = KA*A(1) - A(2)*VM/(V*(A(2)/V + KM))
CP = A(2)/V
DADT(3) = CP
AUCL = A(3) ;AUClast theoretical value

IF (T.EQ.TDOS) THEN
  CMAX = 0
  TMAX = 0
ENDIF

IF (CP.GT.CMAX) THEN
  CMAX = CP
  TMAX = T-TDOS
ENDIF

; Nonlinear pk
;A very large last time needs to be set for AUCinf (Note: different
from the last time of the study design, where AUClast is measured)

$ERROR
IPRED = A(2)/V
IRES = DV-IPRED
PROP = SQRT(SIGMA(1,1))*IPRED
W = PROP
IF (W.EQ.0) W=1
IWRES = IRES/W
Y = IPRED + IPRED*EPS(1)

```

\$THETA

(0,46) ; 1.VM  
(0,400) ; 2.V  
1 FIX ; 3.F  
(0, 1.2) ; 4.KA  
(0,1.25) ; 5.FTRT  
(0, 1) ; 6. KATRT  
(0, 1) ; 7.FSEQ  
(0, 1) ; 8. KASEQ  
(0, 1) ; 9.FPER  
(0, 1) ; 10. KAPER  
(0, 5) ; 11. KM

\$OMEGA

0.03 ; 1. VM  
0.02 ; 2. V  
0.03 ; 3. KA  
\$OMEGA BLOCK(1) 0.0025 ;IOVF  
\$OMEGA BLOCK(1) SAME ;IOVF

\$SIGMA

0 FIX ;PROP

\$SIMULATION (12345) ONLYSIM NSUBPROBS=1

\$TABLE ID TIME TAD CP AUCI AUCL CMAX TMAX  
ONEHEADER NOPRINT FILE=sim\_result.txt